
1. Overview

This document lists TTCN changes needed to correct issues in the ATS ‘iwd-TTCN3-B2016-06_D17wk49’.

3. Corrections to TC 8.5.4.1
Change 1 – Correction to function ‘f_CheckUE_SupportedEUTRA_BandsExt’
	Function name
	f_CheckUE_SupportedEUTRA_BandsExt ()

	Reason for change
	It is assumed that “rf_Parameters_v9e0.supportedBandListEUTRA_v9e0” will always be reported. But in case that no external bands are supported, this parameter will not be reported.

	Summary of change
	The check of parameter “rf_Parameters_v9e0.supportedBandListEUTRA_v9e0” was removed if no external bands are supported.

	TTCN module
	EUTRA_CapabilityFunctions.ttcn

	MCC160 Comment
	


Before change
...
 function f_CheckUE_SupportedEUTRA_BandsExt(UE_EUTRA_Capability p_ReceivedCapabilityMsg)

  {

    // @sic R5-162652, R5-163090, R5-168397 sic@

    var UE_EUTRA_Capability v_ReceivedCapabilityMsg := p_ReceivedCapabilityMsg;

    var bitstring v_SuppEutraBandsExtAccToPICS := f_DeriveSuppEutraBandsFromPics_v9e0();

    var bitstring v_SuppEutraBandsExtAccToPICS_Chunk;

    var integer v_Result;

    var UE_EUTRA_Capability_v9a0_IEs v_Ue_EUTRA_Capability_v9a0_IEs;

    var bitstring v_Ue_EUTRA_Capability_v9a0_IEsContainer;

    var SupportedBandListEUTRA_v9e0 v_SupportedBandListEUTRA_v9e0;

    if (isvalue(v_ReceivedCapabilityMsg.nonCriticalExtension.nonCriticalExtension.lateNonCriticalExtension)) { // @sic R5s120623 sic@ @sic R5s130863 sic@

      // Rel 9 (UE-EUTRA-Capability-v9a0-IEs) IEs  provided

      // decoding of the UE_EUTRA_Capability_v9a0_IEs CONTAINER

      v_Ue_EUTRA_Capability_v9a0_IEsContainer := oct2bit(v_ReceivedCapabilityMsg.nonCriticalExtension.nonCriticalExtension.lateNonCriticalExtension);

      v_Result := decvalue(v_Ue_EUTRA_Capability_v9a0_IEsContainer, v_Ue_EUTRA_Capability_v9a0_IEs);

      if (v_Result != 0) {

        FatalError(__FILE__, __LINE__, "UE_EUTRA_Capability_v9a0_IEs cannot be decoded");

      }

    }

    else {

      FatalError(__FILE__, __LINE__, "UE_EUTRA_Capability_v9a0_IEs not provided");

    }

    v_SupportedBandListEUTRA_v9e0 := v_Ue_EUTRA_Capability_v9a0_IEs.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.rf_Parameters_v9e0.supportedBandListEUTRA_v9e0;

    // It is checked that number of supported EUTRA bands match the set PICS parameters (pc_eBanda_Supp)

    // There may be more bands reported than there are PICS defined for bands

    // Check that any value B such that pc_eBandB_Supp is TRUE and different from all eutra-Band[k] where k = 1 to a - 1

    // Go through the received supportedBandListEUTRA and match against the supported bands according to the PICS

    // Check only the bands for which there are PICS; do not check other bands which may or may not be reported.

    // Done for all sets of continuous PICS.

    v_SuppEutraBandsExtAccToPICS_Chunk := substr(v_SuppEutraBandsExtAccToPICS, 0, 2); // bands 65-66

    f_CheckUE_SupportedEUTRA_BandsExt_Chunk(v_SupportedBandListEUTRA_v9e0, v_SuppEutraBandsExtAccToPICS_Chunk, 0, 2); // @sic R5-162652, R5-163090 sic@

    // appearance of band 70

    v_SuppEutraBandsExtAccToPICS_Chunk := substr(v_SuppEutraBandsExtAccToPICS, 5, 1); // band 70

    f_CheckUE_SupportedEUTRA_BandsExt_Chunk(v_SupportedBandListEUTRA_v9e0, v_SuppEutraBandsExtAccToPICS_Chunk, 5, 1); // @sic R5-168397 sic@

  }
...
After change
...
                                                                                                                                         function f_CheckUE_SupportedEUTRA_BandsExt(UE_EUTRA_Capability p_ReceivedCapabilityMsg)

  {

    // @sic R5-162652, R5-163090, R5-168397 sic@

    var UE_EUTRA_Capability v_ReceivedCapabilityMsg := p_ReceivedCapabilityMsg;

    var bitstring v_SuppEutraBandsExtAccToPICS := f_DeriveSuppEutraBandsFromPics_v9e0();

    var boolean v_ExtBandsSupported := false;
    var bitstring v_SuppEutraBandsExtAccToPICS_Chunk;

    var integer v_Result;

    var UE_EUTRA_Capability_v9a0_IEs v_Ue_EUTRA_Capability_v9a0_IEs;

    var bitstring v_Ue_EUTRA_Capability_v9a0_IEsContainer;

    var SupportedBandListEUTRA_v9e0 v_SupportedBandListEUTRA_v9e0;

    var integer i := 0;
    if (isvalue(v_ReceivedCapabilityMsg.nonCriticalExtension.nonCriticalExtension.lateNonCriticalExtension)) { // @sic R5s120623 sic@ @sic R5s130863 sic@

      // Rel 9 (UE-EUTRA-Capability-v9a0-IEs) IEs  provided

      // decoding of the UE_EUTRA_Capability_v9a0_IEs CONTAINER

      v_Ue_EUTRA_Capability_v9a0_IEsContainer := oct2bit(v_ReceivedCapabilityMsg.nonCriticalExtension.nonCriticalExtension.lateNonCriticalExtension);

      v_Result := decvalue(v_Ue_EUTRA_Capability_v9a0_IEsContainer, v_Ue_EUTRA_Capability_v9a0_IEs);

      if (v_Result != 0) {

        FatalError(__FILE__, __LINE__, "UE_EUTRA_Capability_v9a0_IEs cannot be decoded");

      }

    }

    else {

      FatalError(__FILE__, __LINE__, "UE_EUTRA_Capability_v9a0_IEs not provided");

    }

    // new bands may be specified ...

    while (i < lengthof(v_SuppEutraBandsExtAccToPICS) and (v_ExtBandsSupported == false)) {

      if (v_SuppEutraBandsExtAccToPICS[i] == '1'B) {

        v_ExtBandsSupported := true;

      }

      i := i + 1;

    }
    if (v_ExtBandsSupported == true){

      if (not isvalue (v_Ue_EUTRA_Capability_v9a0_IEs.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.rf_Parameters_v9e0.supportedBandListEUTRA_v9e0)) {

        f_SetVerdict(fail, __FILE__, __LINE__, "Extended bands are claimed to be supported but not reported");         

      }

      else {
        v_SupportedBandListEUTRA_v9e0 := v_Ue_EUTRA_Capability_v9a0_IEs.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.rf_Parameters_v9e0.supportedBandListEUTRA_v9e0;      

        // It is checked that number of supported EUTRA bands match the set PICS parameters (pc_eBanda_Supp)

        // There may be more bands reported than there are PICS defined for bands

        // Check that any value B such that pc_eBandB_Supp is TRUE and different from all eutra-Band[k] where k = 1 to a - 1

        // Go through the received supportedBandListEUTRA and match against the supported bands according to the PICS

        // Check only the bands for which there are PICS; do not check other bands which may or may not be reported.

        // Done for all sets of continuous PICS.

        v_SuppEutraBandsExtAccToPICS_Chunk := substr(v_SuppEutraBandsExtAccToPICS, 0, 2); // bands 65-66

        f_CheckUE_SupportedEUTRA_BandsExt_Chunk(v_SupportedBandListEUTRA_v9e0, v_SuppEutraBandsExtAccToPICS_Chunk, 0, 2); // @sic R5-162652, R5-163090 sic@

        // appearance of band 70

        v_SuppEutraBandsExtAccToPICS_Chunk := substr(v_SuppEutraBandsExtAccToPICS, 5, 1); // band 70

        f_CheckUE_SupportedEUTRA_BandsExt_Chunk(v_SupportedBandListEUTRA_v9e0, v_SuppEutraBandsExtAccToPICS_Chunk, 5, 1); // @sic R5-168397 sic@

      }

    }      

  }
...
